GSV trick for Whitehead link (VERSIONS 1 and 2)

We want to see whether we can use [GSV, eq. (5.5)] to compute unreduced KR homology for the two-component Whitehead link, aka L5a1. Its linking number is 0, and our results for its n=2 and n=3 KR invariants are as follows:

```
lk = 0; KR2 = Expand[
  (q^(12) + 2 q^8 t^2 + q^8 t + q^6 t^3 + 2 q^6 t^2 + q^4 t^4 + q^2 t^4 + t^5) / (q^8 t^2)];
KR3 = Expand[(q^(18) + q^(16) + q^(14) t^2 + 3 q^(12) t^2 + q^(12) t +
     q^(10) t^3 + 3 q^(10) t^2 + q^(10) t + q^8 t^4 + q^8 t^3 + 2 q^8 t^2 +
     q^6 t^4 + q^4 t^4 + q^2 t^5 + q^2 t^4 + t^5) / (q^(12) t^2)]
```

$$3+\frac{2}{q^4}+\frac{3}{q^2}+q^2+\frac{q^4}{t^2}+\frac{q^6}{t^2}+\frac{1}{t}+\frac{1}{q^2\,t}+\frac{t}{q^4}+\frac{t}{q^2}+\frac{t^2}{q^{10}}+\frac{t^2}{q^8}+\frac{t^2}{q^6}+\frac{t^2}{q^4}+\frac{t^3}{q^{12}}+\frac{t^3}{q^{10}}$$

Thus the "universal" terms Ui := \sum_{Q,s,r\in\Z} D_{Q,s,r} q^{iQ+s} t^r for i=2,3 on the right-hand side of [GSV, eq. (5.5)] are given by (where we leave the parameter a:=alpha undetermined for now):

```
U2 = Expand[(q - 1 / q) (q^(4 lk) KR2 - t^a (q + 1 / q)^2 )]
```

$$-\frac{2}{q^3}+2\,q-\frac{q^3}{t^2}+\frac{q^5}{t^2}-\frac{1}{q\,t}+\frac{q}{t}-\frac{t}{q^3}+\frac{t}{q}-\frac{t^2}{q^7}+\frac{t^2}{q^3}-\frac{t^3}{q^9}+\frac{t^3}{q^7}+\frac{t^a}{q^3}+\frac{t^a}{q}-q\,t^a-q^3\,t^a$$

```
U3 = Expand[Simplify[(q - 1 / q) (q^(6 lk) KR3 - t^a ((q^3 - q^(-3)) / (q - 1 / q))^2 )]]
```

$$-\frac{2}{q^5}-\frac{1}{q^3}+2\,q+q^3-\frac{q^3}{t^2}+\frac{q^7}{t^2}-\frac{1}{q^3\,t}+\frac{q}{t}-\frac{t}{q^5}+\frac{t}{q}-\frac{t^2}{q^{11}}+\frac{t^2}{q^3}-\frac{t^3}{q^{13}}+\frac{t^3}{q^9}+\frac{t^a}{q^5}+\frac{t^a}{q^3}+\frac{t^a}{q}-q\,t^a-q^3\,t^a-q^5\,t^a$$

Now we can try to use [GSV, eq. (5.5)] to reproduce our n=3 KR invariant (note that the two components involved are just two unknots by themselves). It works:

```
KR3 - Expand[
  Simplify[q^(-6 lk) (t^a Simplify[((q^3 - q^(-3)) / (q - 1 / q))^2] + U3 / (q - 1 / q) )]]
```

$$0$$

Now try to determine the constants D_{Q,s,r} from U2 and U3 above. If we set a=alpha=0, then U2 and U3 have the same number of terms:

```
U2 /. {a → 0}
```

$$-\frac{1}{q^3}+\frac{1}{q}+q-q^3-\frac{q^3}{t^2}+\frac{q^5}{t^2}-\frac{1}{q\,t}+\frac{q}{t}-\frac{t}{q^3}+\frac{t}{q}-\frac{t^2}{q^7}+\frac{t^2}{q^3}-\frac{t^3}{q^9}+\frac{t^3}{q^7}$$

```
U3 /. {a → 0}
```

$$-\frac{1}{q^5}+\frac{1}{q}+q-q^5-\frac{q^3}{t^2}+\frac{q^7}{t^2}-\frac{1}{q^3\,t}+\frac{q}{t}-\frac{t}{q^5}+\frac{t}{q}-\frac{t^2}{q^{11}}+\frac{t^2}{q^3}-\frac{t^3}{q^{13}}+\frac{t^3}{q^9}$$

From this we can read off the constants D_{Q,s,r}:

```
UU[n_] := - q^(-2 n + 1) + q^(0 n - 1) + q - q^(2 n - 1) - q^(0 n + 3) t^(-2) +
  q^(2 n + 1) t^(-2) - q^(-2 n + 3) t^(-1) + q^(1) t^(-1) - q^(-2 n + 1) t + q^(-1) t -
  q^(-4 n + 1) t^2 + q^(-3) t^2 - q^(-4 n - 1) t^3 + q^(-2 n - 3) t^3; UU[n]
```

$$\frac{1}{q}+q-q^{1-2\,n}-q^{-1+2\,n}-\frac{q^3}{t^2}+\frac{q^{1+2\,n}}{t^2}+\frac{q}{t}-\frac{q^{3-2\,n}}{t}+\frac{t}{q}-q^{1-2\,n}\,t+\frac{t^2}{q^3}-q^{1-4\,n}\,t^2-q^{-1-4\,n}\,t^3+q^{-3-2\,n}\,t^3$$

Now we use [GSV, eq. (5.5)] with a=alpha=0 to define a candidate for the KR invariants for arbitrary n:

```
KR[n_] := Expand[FullSimplify[Expand[Simplify[(q^(-2 n lk)
        (t^a Simplify[((q^n - q^(-n)) / (q - 1 / q))^2] + UU[n] / (q - 1 / q) )) /. {a → 0}]]]]
```

Check that we have not made a mistake so far:

```
{KR[2] - KR2, KR[3] - KR3}
```

```
{0, 0}
```

Our result for the n=4 invariant is:

```
KR4 = Expand[(q^(24) + q^(22) + q^(20) t^2 + q^(20) + 2 q^(18) t^2 + 4 q^(16) t^2 +
      q^(16) t + q^(14) t^3 + 4 q^(14) t^2 + q^(14) t + q^(12) t^4 + q^(12) t^3 +
      3 q^(12) t^2 + q^(12) t + q^(10) t^4 + q^(10) t^3 + 2 q^(10) t^2 + q^8 t^4 +
      q^6 t^4 + q^4 t^5 + q^4 t^4 + q^2 t^5 + q^2 t^4 + t^5) / (q^(16) t^2)];
```

They agree with the GSV prediction:

```
KR[4] - KR4
```

```
0
```

The general expression for the GSV predicion is:

```
FullSimplify[
 (q^(-2 n lk) ( t^a Simplify[((q^n - q^(-n)) / (q - 1 / q))^2] + UU[n] / (q - 1 / q) )) /. {a → 0}]
```

$$\frac{q^{-2-4n}\left(q^{4+2n}\left(-1+q^{2n}\right)^2+\frac{\left(-1+q^2\right)\left(-q^2+q^{2n}\right)\left(q^{2+4n}(q-t)(q+t)+t^4\left(q^2+t\right)+q^{2n}t\left(q^2+t\right)\left(q^2+t^2\right)\right)}{t^2}\right)}{\left(-1+q^2\right)^2}$$