We want to see whether we can use [GSV, eq. (5.5)] to compute unreduced KR homology for the two-component link "Solomon's knot", aka L4a1. Its linking number is 2, and our results for its n=2 and n=3 KR invariants are as follows:

```
In[1]:= lk = 2; KR2 = Expand[(t^4 q^2 + t^4 + t^2 q^4 + t q^8 + q^(10) + q^8) / q^(10)];
        KR3 = Expand[(q^(16) + q^(14) t + q^(14) + q^(12) t + q^(12) +
             q^(8) t^2 + q^6 t^4 + q^6 t^2 + 2 q^4 t^4 + 2 q^2 t^4 + t^4) / (q^(16))]
```

Out[1]= $1+\frac{1}{q^4}+\frac{1}{q^2}+\frac{t}{q^4}+\frac{t}{q^2}+\frac{t^2}{q^{10}}+\frac{t^2}{q^8}+\frac{t^4}{q^{16}}+\frac{2\,t^4}{q^{14}}+\frac{2\,t^4}{q^{12}}+\frac{t^4}{q^{10}}$

Thus the "universal" terms Ui := \sum_{Q,s,r\in\Z} D_{Q,s,r} q^{iQ+s} t^r for i=2,3 on the right-hand side of [GSV, eq. (5.5)] are given by (where we leave the parameter a:=alpha undetermined for now):

```
In[2]:= U2 = Expand[(q - 1 / q) (q^(4 lk) KR2 - t^a (q + 1 / q)^2)]
```

Out[2]= $-q^5+q^9-q^5\,t+q^7\,t-q\,t^2+q^3\,t^2-\frac{t^4}{q^3}+q\,t^4+\frac{t^a}{q^3}+\frac{t^a}{q}-q\,t^a-q^3\,t^a$

```
In[3]:= U3 = Expand[Simplify[(q - 1 / q) (q^(6 lk) KR3 - t^a ((q^3 - q^(-3)) / (q - 1 / q))^2)]]
```

Out[3]= $-q^7+q^{13}-q^7\,t+q^{11}\,t-q\,t^2+q^5\,t^2-\frac{t^4}{q^5}-\frac{t^4}{q^3}+q\,t^4+q^3\,t^4+\frac{t^a}{q^5}+\frac{t^a}{q^3}+\frac{t^a}{q}-q\,t^a-q^3\,t^a-q^5\,t^a$

Now we can try to use [GSV, eq. (5.5)] to reproduce our n=3 KR invariant (note that the two components involved are just two unknots by themselves). It works:

```
In[4]:= Expand[Simplify[q^(-6 lk) (t^a Simplify[((q^3 - q^(-3)) / (q - 1 / q))^2] + U3 / (q - 1 / q))]]
```

Out[4]= $1+\frac{1}{q^4}+\frac{1}{q^2}+\frac{t}{q^4}+\frac{t}{q^2}+\frac{t^2}{q^{10}}+\frac{t^2}{q^8}+\frac{t^4}{q^{16}}+\frac{2\,t^4}{q^{14}}+\frac{2\,t^4}{q^{12}}+\frac{t^4}{q^{10}}$

Now try to determine the constants D_{Q,s,r} from U2 and U3 above. If we set a=alpha=4, then U2 and U3 have the same number of terms:

```
In[5]:= U2 /. {a → 4}
```

Out[5]= $-q^5+q^9-q^5\,t+q^7\,t-q\,t^2+q^3\,t^2+\frac{t^4}{q}-q^3\,t^4$

```
In[6]:= U3 /. {a → 4}
```

Out[6]= $-q^7+q^{13}-q^7\,t+q^{11}\,t-q\,t^2+q^5\,t^2+\frac{t^4}{q}-q^5\,t^4$

From this we can read off the constants D_{Q,s,r}:

```
In[7]:= UU[n_] := - q^(2 n + 1) + q^(4 n + 1) - q^(2 n + 1) t + q^(4 n - 1) t -
          q^(0 n + 1) t^2 + q^(2 n - 1) t^2 + q^(0 n - 1) t^4 - q^(2 n - 1) t^4; UU[m]
```

Out[7]= $-q^{1+2\,m}+q^{1+4\,m}-q^{1+2\,m}\,t+q^{-1+4\,m}\,t-q\,t^2+q^{-1+2\,m}\,t^2+\frac{t^4}{q}-q^{-1+2\,m}\,t^4$

Now we use [GSV, eq. (5.5)] with a=alpha=4 to define a candidate for the KR invariants for arbitrary n:

```
In[8]:= KR[n_] := Expand[FullSimplify[Expand[Simplify[(q^(-2 n lk)
              (t^a Simplify[((q^n - q^(-n)) / (q - 1 / q))^2] + UU[n] / (q - 1 / q))) /. {a → 4}]]]]
```

Check that we have not made a mistake so far:

In[9]:=
```
{KR[2] - KR2, KR[3] - KR3}
```

Out[9]=
```
{0, 0}
```

Our results for the n=4 and n=5 KR invariants are:

In[10]:=
```
KR4 = Expand[(q^(22) + q^(20) t + q^(20) + q^(18) t +
      q^(18) + q^(16) t + q^(16) + q^(12) t^2 + q^(10) t^4 + q^(10) t^2 +
      2 q^8 t^4 + q^8 t^2 + 3 q^6 t^4 + 3 q^4 t^4 + 2 q^2 t^4 + t^4) / (q^(22))];
KR5 = Expand[(q^(28) + q^(26) t + q^(26) + q^(24) t + q^(24) + q^(22) t + q^(22) + q^(20) t +
      q^(20) + q^(16) t^2 + q^(14) t^4 + q^(14) t^2 + 2 q^(12) t^4 + q^(12) t^2 + 3 q^(10)
       t^4 + q^(10) t^2 + 4 q^8 t^4 + 4 q^6 t^4 + 3 q^4 t^4 + 2 q^2 t^4 + t^4) / (q^(28))];
```

They agree with the GSV prediction:

In[12]:=
```
{KR[4] - KR4, KR[5] - KR5}
```

Out[12]=
```
{0, 0}
```

The general expression for the GSV predicion is:

In[13]:=
```
FullSimplify[
 (q^(-2 n lk) (t^a Simplify[((q^n - q^(-n)) / (q - 1 / q))^2] + UU[n] / (q - 1 / q) )) /. {a → 4}]
```

Out[13]=
$$\frac{q^{-4n}\left(q^{2-2n}\left(-1+q^{2n}\right)^2 t^4+\left(-1+q^2\right)\left(-q^2 t^2+t^4+q^{4n}\left(q^2+t\right)-q^{2n}(1+t)\left(q^2+(-1+t)\,t^2\right)\right)\right)}{\left(-1+q^2\right)^2}$$